CRITERION FOR SAFE LOADS ON ICE SHEETS

54-21-17-40
Col.

Report No. 742G-72-11
September, 1972

by

William L. Hill

SUNOCO E & P LIMITED
CALGARY ALBERTA

ABSTRACT

The safe load criterion predicts the maximum load capable of being supported by a given thickness of ice as a function of the ratio of load radius to ice thickness. The mathematical approach is based on the critical stress created by distributed loads on plates. The criterion was evaluated using the engineering ice data from the ice mechanics program conducted at Yellowknife, N.W.T.

CRITERION FOR SAFE LOADS ON ICE SHEETS

SAFE LOAD CONDITIONS CAN BE PREDICTED

A safe load criterion for ice sheets has been established by extending Nevel's mathematical study [1] of concentrated loads on plates. The engineering ice data from the ice mechanics program at Yellowknife, N.W.T., was used to evaluate this criterion. The criterion can predict the maximum load that may be supported by an ice sheet of a given thickness.

The safe load criterion predicts the safe load conditions encountered during the Yellowknife test. A 194-ton air cushion vehicle was tested on the ice sheet at Yellowknife. The various deck loads and ice thicknesses were noted for failure and non-failure of the ice sheet. These conditions are predictable by the safe load criterion. Another set of datum points used to check the criterion was the loading conditions for four static load tests [2]. Once again the safe load conditions are predicted by the criterion.

Accurate values for the engineering ice properties are essential to the reliability of this criterion. Field tests have confirmed its application to lake ice as encountered at Yellowknife. Application to other types of fresh water ice and sea ice should be made after testing the criterion against known loading conditions.

APPLICATION OF THE SAFE LOAD CRITERION

The criterion determines the maximum load that may be placed on a given thickness of ice as a function of the ratio of load radius to ice thickness. The mathematical basis of the criterion is discussed in Appendix A. A computer program was written to analyze specified loading conditions. A user's guide for the program is presented in Appendix B.

The computer program calculates the critical safety factor and compares it with the actual safety factor. The ice sheet will support the load when the actual safety factor is less than the critical safety factor. The computer program requires the following input data: load (lbs.), load radius (ft.), ice thicknesses (ft.), Poisson's ratio of the ice (non-dimensional), elastic modulus of the ice (psf), and the failure strength of the ice (psf).

Table I shows the excellent agreement between the safe load criterion and the actual responses of the Yellowknife ice sheet for various loads, load radii, and ice thicknesses.

A family of curves based on ice thicknesses can be generated to represent the safe load as a function of the load radius divided by the ice thickness. The curves plotted in Figure 1 were generated using the engineering properties of the Yellowknife ice sheet. The loading conditions given in Table I are plotted in Figure 1. The circles and the squares are the ACT-100 tests and the static load tests, respectively. Open symbols indicate that the ice supported the load and solid symbols indicate that the ice failed.

TABLE I

SAFE LOAD CRITERION APPLIED TO YELLOWKNIFE TESTS

TESTS	ICE THICKNESS (inches)	LOAD (lbs)	RATIO OF LOAD RADIUS TO ICE THICKNESS	SAFE LOAD CRITERION	SAFE LOAD TEST
ACT-100					
No. 1	12	388,000	33.6	No	No
No. 2	16	588,000	25.2	No	No
No. 3	22	588,000	18.3	No	No
No. 4	22	388,000	18.3	Yes	Yes
No. 5	27	588,000	14.9	No	No
No. 6	27	488,000	14.9	Yes	Yes
No. 7	37	588,000	10.9	Yes	Yes
Static Load					
No. 1	13	43,103	8.53	Yes	Yes
No. 2	15.5	61,575	7.14	Yes	Yes
No. 3	19	152,400	8.57	Yes	Yes
No. 4	40	388,000	10.10	Yes	Yes



FIGURE 1 SAFE LOAD CRITERION FOR FRESH WATER ICE

RECOMMENDATIONS

1. The safe load criterion is recommended for use on fresh water ice.

2. Additional tests are recommended to confirm its validity for sea ice. Field tests are necessary to develop confidence level high enough to risk lives and equipment on its predictions of safe load conditions.

3. Tests in model ice basins are recommended only as supplemental tests to the field tests.

4. The recommended extension of this work is to develop a criterion to optimize the safe load spacing between multiple loads placed on an ice sheet. Optimization of the spacing will permit us to place more camp materiel on the ice sooner and stay longer.

REFERENCES

1. Nevel, D. E., Concentrated Loads on Plates, CRREL Research Report No. 265 (March, 1970) 11p.

2. Hill, W. L., Static Load Tests on the Yellowknife Ice Sheet, Sun Oil Company, Production Research Department, Report No. 742G-72-7 (July, 1972).

APPENDIX A
MATHEMATICAL BASIS FOR THE SAFE LOAD CRITERION

MATHEMATICAL BASIS FOR THE SAFE LOAD CRITERION

Nevel [1] presented a mathematical solution for the stress occurring in an ice sheet subjected to a load distributed uniformly over a circular area. He assumed the ice sheet to be a homogeneous, isotropic, elastic plate resting on an elastic foundation. Nevel showed that Love's method using a stress function and Nadai's method, using stress in terms of displacement will both lead to the same result.

The general solution for the stresses (σ_z, σ_θ, τ_{rz}) and the displacement (u-radial, w-vertical) are given by Nevel. The stress under the applied load is obtained by evaluating the general solution at the underneath surface of the ice sheet ($r = o$, $z = h$). The resulting mathematical solution takes the following form:

$$\frac{\sigma h^2}{P(1+\mu)} = \frac{1}{\pi} \int_0^\infty \frac{12t^2 (1-e^{-2t})}{DEN} e^{-t} \frac{J_1 (t\beta)}{\beta} dt$$

$$+ \frac{1}{\pi} \int_0^\infty \left(\frac{h}{\ell}\right)^4 \left[\frac{t-1 + (t+1)e^{-2t}}{DEN}\right] e^{-t} \frac{J_1 (t\beta)}{\beta} dt$$

$$- \frac{\mu}{1+\mu} \frac{1}{\pi} \int_0^\infty \left(\frac{h}{\ell}\right)^4 \left[\frac{t+1+(t-1)e^{-2t}}{DEN}\right] e^{-t} \frac{J_1(t\beta)}{\beta} dt \qquad (1)$$

where

$$DEN = \left[6t + \left(\frac{h}{\ell}\right)^4\right] + 4t \left[\left(\frac{h}{\ell}\right)^4 - 3 - 6t^2\right] e^{-2t}$$

$$+ \left[6t - \left(\frac{h}{\ell}\right)^4\right] e^{-4t} \qquad (2)$$

$$\ell = \sqrt[4]{\frac{Eh^3}{12\rho(1-\mu^2)}}$$

$$\beta = b/h \qquad (3)$$

σ = stress at r=o, h=z, psf

ρ = density of ice, pcf

μ = Poisson's ratio of ice, non-dimensional

E = Elastic modulus of ice, psf

h = Ice thickness, ft.

P = Load on ice, lbs.

b = Load radius, ft.

t = transform variable, non-dimensional

A safety factor is defined by Equation (5).

$$S.F. = \frac{P(1+\mu)}{\sigma h^2} \tag{5}$$

This equation can be evaluated by using Equation (1) provided we know the load radius and the engineering properties of the ice. When Equation (5) is evaluated by using Equation (1), we define the resulting value as the critical safety factor. This critical safety factor will be compared to the actual safety factor.

A value for the actual safety factor is calculated using Equation (5) by substituting the known values for the load on the ice, the Poisson's ratio of the ice, the failure strength of the ice, and the ice thickness.

If the actual safety factor is less than the critical safety factor, the ice will support the load.

APPENDIX B

USER'S GUIDE FOR SAFLOAD:

A COMPUTER PROGRAM FOR THE SAFE LOAD CRITERION

USER'S GUIDE FOR SAFLOAD:

A COMPUTER PROGRAM FOR THE SAFE LOAD CRITERION

SAFLOAD is a computer program that answers the question: "Will the ice sheet h feet thick support a load P distributed over a load radius of b?

The program evaluates the critical safety factor for the ice sheet. The actual safety factor for the load conditions is calculated and compared to the critical safety factor. The ice sheet will support the load when the actual safety factor is less than the critical safety factor.

The input variables are:

XLOAD	Load on the ice, lb.
RLOAD	Radius of load, ft.
EMOD	Elastic Modulus, psf
PRATIO	Poisson's ratio, non-dimensional
WDEN	Water density, lb/ft^3
TICE	Thickness of ice, ft.
FSTREN	Failure strength of ice, psf
NCASE	Number of cases for XLOAD and RLOAD
NICE	Number of cases for TICE

The data cards are ordered in the following manner:

Card No.	Format	Variables
1	2E12.4, 2F10.4	EMOD, FSTREN, WDEN, PRATIO
2	I2, I2	NCASE, NICE
3	E12.4, F10.4	XLOAD, RLOAD
4	F10.4	TICE

Example Problem

Input data cards —

Card No.	Input Data
1	1.866E+08, 6.062E+04, 62.4,0.333
2	4,4
3	4.3103E+04, 9.21
4	1.08
5	1.29
6	1.58
7	3.33
8	6.1575E+04, 9.21
9	1.08
10	1.29
11	1.58
12	3.33
13	1.5240E+05, 13.54
14	1.08
15	1.29
16	1.58
17	3.33
18	3.8800E+05, 33.57
19	1.08
20	1.29
21	1.58
22	3.33

Computer Output:

SAFE LOAD CRITERION FOR ICE SHEETS

ICE THICKNESS (FT)	LOAD ON ICE (LBS)	LOAD RADIUS (FT)	SAFETY FACTORS ACTUAL	SAFETY FACTORS CRITICAL	LOAD SAFE
1,0800	4.3103E+04	9.2100	,8127	1.3059	YES
1,2900	4.3103E+04	9.2100	,5697	1.2077	YES
1,5800	4.3103E+04	9.2100	,3797	1.1115	YES
3,3300	4.3103E+04	9.2100	,0855	.8567	YES
1,0800	6.1575E+04	9.2100	1,1610	1.3059	YES
1,2900	6.1575E+04	9.2100	,8138	1.2077	YES
1,5800	6.1575E+04	9.2100	,5425	1.1115	YES
3,3300	6.1575E+04	9.2100	,1221	.8567	YES
1,0800	1.5240E+05	13.5400	2,8736	1.6982	NO
1,2900	1.5240E+05	13.5400	2,0141	1.5399	NO
1,5800	1.5240E+05	13.5400	1,3426	1.3901	YES
3,3300	1.5240E+05	13.5400	,3023	1.0178	YES
1,0800	3.8800E+05	33.5700	7,3159	4.6115	NO
1,2900	3.8800E+05	33.5700	5,1279	3.7803	NO
1,5800	3.8800E+05	33.5700	3,4182	3.0984	NO
3,3300	3.8800E+05	33.5700	,7695	1.7844	YES

Listing of SAFLOAD:

```
      PROGRAM SAFLOAD(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
C     WRITTEN BY W.L.WILL MAY 23,1972
C     PROGRAM SAFLOAD CALCULATES A CRITICAL SAFETY FACTOR AND COMPARES
C     IT TO THE ACTUAL SAFETY FACTOR. THE ICE SHEET WILL SUPPORT THE
C     LOAD IF THE ACTUAL SAFETY FACTOR IS LESS THAN THE CRITICAL FACTOR.
C
C     INPUT REQUIRED
C     XLOAD  = LOAD ON THE ICE , LBS
C     RLOAD  = RADIUS OF LOAD   , FT
C     EMOD   = ELASTIC MODULUS , PSF
C     PRATIO = POISSON RATIO    , DIMENSIONLESS
C     WDEN   = WATER DENSITY    , LBS/CU FT
C     TICE   = THICKNESS OF ICE, FT
C     FSTREN = FAILURE STRENGTH OF ICE , PSF
C     NCASE  = NUMBER OF CASES FOR XLOAD AND/OR RLOAD
C     NICE   = NUMBER OF CASES FOR ICE THICKNESS
C
      READ(5,3) EMOD,FSTREN,WDEN,PRATIO
      READ(5,1)NCASE,NICE
1     FORMAT (I2,I2)
      WRITE(6,10)
10    FORMAT(1H1)
      WRITE(6,11)
11    FORMAT(15X,46H      SAFE LOAD CRITERION FOR ICE SHEETS       ,//,1X
     1,74HICE THICKNESS  LOAD ON ICE  LOAD RADIUS    SAFETY FACTORS
     2 LOAD SAFE ,/,4X,4H(FT),10X,5H(LBS),8X,4H(FT),6X,19HACTUAL   CRIT
     3ICAL  ,/)
      DO 90 I=1,NCASE
      WRITE(6,60)
 60   FORMAT(//)
      READ(5,2) XLOAD,RLOAD
2     FORMAT (E12.4,F10.4)
3     FORMAT(2E12.4,2F10.4)
      DO 91 J=1,NICE
      READ(5,4) TICE
4     FORMAT(F10.4)
      PRATIO2 = PRATIO**2.
      TICE3 = TICE**3.
      DUM = 12.*WDEN*(1. - PRATIO2)
      XL4 = EMOD*TICE3 / DUM
      XL = XL4**(1./4.)
      ASTRESS = (FSTREN*(TICE)**2)/(XLOAD*(1. + PRATIO))
      CALL STRESF(RLOAD,TICE,XL,TSTRESS)
      IF (ASTRESS.LE.TSTRESS) GO TO 70
      ANS = 3HYES
      GO TO 71
70    ANS = 3H NO
71    CONTINUE
      ASF = 1. / ASTRESS
      CSF = 1. / TSTRESS
      WRITE(6,20) TICE,XLOAD,RLOAD,ASF,CSF,ANS
20    FORMAT(F10.4,4X,E12.4,2X,F10.4,3X,F6.4,4X,F6.4,9X,A3)
91    CONTINUE
90    CONTINUE
      END
```

```
      SUBROUTINE STRESF(B,H,XL,SF)
C
      DIMENSION F(2010)
      NPT = 2001
      D = 0.0001
      BETA = B/H
      HOL = H/XL
      HOL4 = HOL**4
      XNPT = NPT
      X = 10./(XNPT-1.0)
      DO 3 K=1,NPT
      XK = K
      T = X*(XK-1.0)
      DEN = ((6.0*T)+HOL4)+(4.0*T)*((HOL4-3.0-(6.0*T*T))*EXP(-2.0*T))
     1+((6.0*T)-HOL4)*EXP(-4.0*T)
      TBETA = T*BETA
      CALL BESJ(TBETA,1,BJ,D,IER)
      IF((BETA.EQ.0.0).OR.(DEN.EQ.0.0)) GO TO 9
      F(K)=(12.0*T*T*(1.0-EXP(-2.0*T))/DEN)*EXP(-T)*(BJ/BETA)
      GO TO 3
    9 F(K) = 0.0
    3 CONTINUE
      F1 = F(1)+F(NPT)
      F2 = 0.0
      NM1 = NPT-1
      DO 4 L=2,NM1,2
      F2 = F2+F(L)
    4 CONTINUE
      F2 = F2*4.0
      F3 = 0.0
      NM2 = NPT-2
      DO 6 L=3,NM2,2
      F3 = F3+F(L)
    6 CONTINUE
      F3 = F3*2.0
      SF = ((1.0/3.0)*X*(F1+F2+F3))/3.141593
      RETURN
      END


      SUBROUTINE BESJ(X,N,BJ,D,IER)
C
      BJ=.0
      IF(N)10,20,20
   10 IER=1
      RETURN
   20 IF(X)30,30,31
   30 IER=2
      RETURN
   31 IF(X-15.)32,32,34
   32 NTEST=20.+10.*X-X** 2/3
      GO TO 36
   34 NTEST=90.+X/2.
   36 IF(N-NTEST)40,38,38
   38 IER=4
      RETURN
```

```
   40 IER=0
      N1=N+1
      BPREV=.0
C
C     COMPUTE STARTING VALUE OF M
C
      IF(X-5.)50,60,60
   50 MA=X+6.
      GO TO 70
   60 MA=1.4*X+60./X
   70 MB=N+IFIX(X)/4+2
      MZERO=MAX0(MA,MB)
C
C     SET UPPER LIMIT OF M
C
      MMAX=NTEST
  100 DO 190 M=MZERO,MMAX,3
C
C     SET F(M),F(M-1)
C
      FM1=1.0E-28
      FM=.0
      ALPHA=.0
      IF(M-(M/2)*2)120,110,120
  110 JT=-1
      GO TO 130
  120 JT=1
  130 M2=M-2
      DO 160 K=1,M2
      MK=M-K
      BMK=2.*FLOAT(MK)*FM1/X-FM
      FM=FM1
      FM1=BMK
      IF(MK-N-1)150,140,150
  140 BJ=BMK
  150 JT=-JT
      S=1+JT
  160 ALPHA=ALPHA+BMK*S
      BMK=2.*FM1/X-FM
      IF(N)180,170,180
  170 BJ=BMK
  180 ALPHA=ALPHA+BMK
      BJ=BJ/ALPHA
      IF(ABS(BJ-BPREV)-ABS(D*BJ))200,200,190
  190 BPREV=BJ
      IER=3
  200 RETURN
      STOP
      END
```